Abstract

SAS/STAT® 9.3, coming soon to a site near you, delivers numerous enhancements to the statistical software. The PHREG procedure supports frailty models for incorporating random effects in Cox regression, and the MCMC procedure provides a RANDOM statement to facilitate fitting Bayesian models with random effects. The NLIN procedure has been updated, and the MI procedure offers additional flexibility by providing a fully conditional specification method. The new SURVEYPHREG and HPMIXED procedures are also outfitted with additional capabilities.

This talk reviews the highlights of SAS/STAT 9.22 and then describes important SAS/STAT 9.3 enhancements with practical illustrations, mainly from the SAS/STAT 9.3 documentation.

Introduction

SAS/STAT has undergone considerable updates in recent years: SAS/STAT 9.2 brought major enhancements to the product: Bayesian capabilities in the GENMOD, LIFEREG, and PHREG procedures; the MCMC procedure for fitting general Bayesian models; the experimental HPMIXED procedure for performing mixed model analysis with a large number of observations and/or class levels; the SEQDESIGN and SEQTEST procedures for group sequential analysis; and production ODS Statistical Graphics. In addition, more than 200 new features were added to the existing software.

In 2009, SAS/STAT 9.22 arrived. It made a full complement of postfitting capabilities available in a large number of linear modeling procedures as well as introducing the PLM procedure, which enables you to take stored model information and use it to perform additional inference and scoring without refitting the original model. The SURVEYPHREG procedure provides survival analysis, in the form of Cox proportional hazards regression, for sample survey data. The more powerful and customizable structural equation modeling first implemented with the experimental TCALIS procedure in SAS/STAT 9.2 was rolled into the CALIS procedure. Other enhancements include exact Poisson regression, zero-inflated negative binomial models, model-averaging for high-dimension prediction, and improved prediction of spatial processes with the spatial analysis procedures.

SAS/STAT 9.3, available later this year, both introduces new methodology as well as continues to refine existing functionality. The SURVEYPHREG procedure becomes production and now handles time-dependent covariates. The EFFECT statement also becomes production and is available in eleven linear modeling procedures. The MCMC procedure provides a RANDOM statement, which simplifies the specification of hierarchical random-effects models and significantly reduces simulation time which improving convergence. This statement defines random effects that can enter the model in a linear or nonlinear fashion and supports univariate and multivariate prior distributions.

The experimental FMM procedure fits statistical models to data where the distribution of the response is a finite mixture of univariate distributions. These models are useful for applications such as estimating multimodal or heavy-tailed densities, fitting zero-inflated or hurdle models to count data with excess zeros, modeling overdispersed data, and fitting regression models with complex error distributions. The MI procedure now includes the experimental FCS statement, which specifies a multivariate imputation by fully conditional specification (FCS) methods. Finally, The NLIN procedure has been updated with experimental features for diagnosing the nonlinear model fit.

This paper reviews some of the new capabilities in the SAS/STAT 9.3 release. It draws heavily from the documentation in progress for this release.

A Review of SAS/STAT 9.22

Many users are still finding out about SAS/STAT 9.22, a new release that ‘snuck out’ with the third maintenance of SAS® 9.3. This release contained many important enhancements to SAS/STAT software.

New architectural changes mean that many procedures added additional postfitting capabilities to their arsenal. Over 30 postfitting statements have been added to the linear modeling procedures. In addition, several existing statements have been updated. Table 1 provides an overview of these capabilities. Checks indicate that new statements have been added to procedures, stars indicate existing functionality, and starred checks indicate that existing statements have been updated.
In addition, the PLM procedure performs postfitting inference with model fit information saved from a number of SAS/STAT modeling procedures. These procedures are equipped with the new STORE statement, which saves model information as a SAS item store. An item store is a special SAS binary file that is used to store and restore information that has a hierarchical structure. Ten SAS/STAT procedures now provide the STORE statement: GENMOD, GLIMMIX, GLM, LOGISTIC, MIXED, ORTHOREG, PHREG, SURVEYLOGISTIC, SURVEYPHREG, and SURVEYREG.

The PLM procedure takes these item stores as input and performs tasks such as testing hypotheses, producing effect plots, and scoring a new data set. These tasks are specified through the usual complement of postfitting statements such as the TEST, LSMEANS, and new EFFECTPLOT and SCORE statements. Any procedure that offers the STORE statement can produce the item stores that are necessary for postfitting processing with the PLM procedure. This allows you to perform additional postfitting inference at a later time without having to refit your model, which is especially convenient for those models that are computationally expensive. In addition, with growing concerns for data confidentiality, storing and using intermediate results for remaining analyses might become a requirement in some organizations.

The SURVEYPHREG procedure fits the semiparametric Cox model for proportional odds to sample survey data. Based on work by David Binder (1983), the procedure provides design-based variance estimates, confidence intervals, and hypothesis tests concerning the model parameters and model effects. For statistical inference, PROC SURVEYPHREG incorporates complex survey sample designs, including designs with stratification, clustering, and unequal weighting.

The GENMOD procedure is equipped with an EXACT statement and performs exact Poisson regression as well as exact logistic regression. In addition, it performs zero-inflated negative binomial regression as well as zero-inflated Poisson regression. The EFFECT statement is available in the HPMIXED, GLIMMIX, GLMSELECT, LOGISTIC, ORTHOREG, PHREG, PLS, QUANTREG, ROBUSTREG, SURVEYLOGISTIC, and SURVEYREG procedures. Effect types include splines for semiparametric modeling, effects for situations in which measurements can belong to more than one class, lag effects, and polynomials.


### ODS Statistical Graphics Footprint Grows Larger

The 9.3 release marks another milestone in the evolution of statistical graphics in SAS. The ODS graphical capabilities move to SAS® Base so they will be available to all SAS/STAT customers. This means that a SAS/GRAPH® license will no longer be required to generate the plots produced by the statistical procedures. The SG family of procedures is also part of Base SAS, and so is the ODS Graphics editor and the Graphical Template Language. You might be interested in the ODS Graphics Designer, which enables you to create new graphics via a point-and-click interface. (Note that SAS/GRAPH remains a SAS product with extensive graphics functionality.)

---

<table>
<thead>
<tr>
<th>Table 1</th>
<th>Postfitting Statements Available in Linear Modeling Procedures</th>
</tr>
</thead>
<tbody>
<tr>
<td>Procedure</td>
<td>CONTRAST</td>
</tr>
<tr>
<td>GENMOD</td>
<td>*</td>
</tr>
<tr>
<td>GLM</td>
<td>*</td>
</tr>
<tr>
<td>GLIMMIX</td>
<td>*</td>
</tr>
<tr>
<td>LOGISTIC</td>
<td>*</td>
</tr>
<tr>
<td>MIXED</td>
<td>*</td>
</tr>
<tr>
<td>ORTHOREG</td>
<td>*</td>
</tr>
<tr>
<td>PHREG</td>
<td>*</td>
</tr>
<tr>
<td>SURVEYLOGISTIC</td>
<td>*</td>
</tr>
<tr>
<td>SURVEYPHREG</td>
<td></td>
</tr>
<tr>
<td>SURVEYREG</td>
<td>✓</td>
</tr>
</tbody>
</table>
In addition, the default Display Manager settings have been changed so that resulting tables and graphs are interspersed. ODS Graphics is now automatically turned on in Display Manager, and the default output destination is HTML. The new HTMLBlue style affords a modern appearance. As a result of these changes, a few graphs behave somewhat differently, mostly having to do with the number of points being plotted. See the “What’s Changed” section in the “What’s New” chapter of the SAS/STAT 9.3 documentation for more detail.

In addition, several more graphics have been added to SAS/STAT. Figure 2 displays the agreement plot corresponding to observer agreement analysis in the FREQ procedure:

**Figure 2** Agreement Plot for MS Patient Study
Finite Mixture Models

Finite mixture models allow you to fit statistical models to data when the distribution of the response is a finite mixture of univariate distributions. These models are useful for applications such as estimating multimodal or heavy-tailed densities, fitting zero-inflated or hurdle models to count data with excess zeros, modeling overdispersed data, and fitting regression models with complex error distributions. Many well-known statistical models for dealing with overdispersed data are members of the finite mixture model family: for example, zero-inflated Poisson models and zero-inflated negative binomial models.

The experimental FMM procedure in SAS/STAT 9.3 fits finite mixtures of regression models of generalized linear models in which the regression structure and the covariates are the same or different across the components. PROC FMM performs maximum likelihood estimation for all models, and it provides Markov Chain Monte Carlo estimation for many models, including zero-inflated Poisson models. The procedure includes many built-in link and distribution functions, including the beta, shifted, Weibull, beta-binomial, and generalized Poisson distributions, as well as standard members of the exponential family of distributions. In addition, several specialized built-in mixture models are provided, such as the binomial cluster model (Morel and Nagaraj, 1993).

The FMM procedure uses the typical model-building syntax of SAS statistical procedures, including the CLASS and MODEL statements, although multiple MODEL statements are required to build the mixture models in which model effects, distributions, or link functions vary across mixture components. It also provides a number of features particular to fitting finite mixture models. For example, it evaluates sequences of mixture models when you specify ranges for the number of components, it has the ability to model regression and classification effects in the mixing probabilities, and it can incorporate full or partially known component membership into the analysis. In addition, Bayesian methods are available for several models.

The following data illustrates the use of a zero-inflated model. In a survey of park attendees, randomly selected individuals were asked about the number of fish they caught in the last six months. Along with that count, the gender and age of each sampled individual were recorded. The following DATA step displays the data for the analysis:

data catch;
  input gender $ age count @@;
datalines;
F  54 18   M  37 0   F  48 12  M  27 0
M  55 0   M  32 0   F  49 12  F  45 11
M  39 0   F  34 1   F  50 0   M  52 4
M  33 0   M  32 0   F  23 1   F  17 0
"
If you simply look at the data, it appears that many park attendees did not catch any fish. The zero counts are made up of two populations: attendees who do not fish and attendees who fish badly. A zero-inflation mechanism thus appears reasonable since a zero count can be produced by two separate distributions.

The following PROC FMM statements fit a standard Poisson regression model to these data. A common intercept is assumed for men and women, and the regression slope varies with gender.

```sql
proc fmm data=catch;
  class gender;
  model count = gender*age / dist=Poisson;
run;
```

Figure 4 displays information about the model and data set. The “Model Information” table conveys that the model is a single-component Poisson model (a Poisson GLM) and that parameters are estimated by maximum likelihood.

![Figure 4 Model Information and Class Levels in Poisson Regression](image)

Figure 5 displays the “Fit Statistics” and “Parameter Estimates” tables from the maximum likelihood estimation. If the model is not overdispersed, the Pearson statistic should roughly equal the number of observations in the data set minus the number of parameters. With \( n = 52 \), there is evidence of overdispersion in these data with \( Q_P = 85.96 \).

![Figure 5 Fit Results in Poisson Regression](image)
Suppose that the cause of overdispersion is zero-inflation of the count data. The following statements fit a zero-inflated Poisson model.

```plaintext
proc fmm data=catch;
   class gender;
   model count = gender*age / dist=Poisson ;
   model + / dist=Constant;
run;
```

There are two MODEL statements, one for each component of the mixture. The distributions are different for the components, so you cannot specify the mixture model with a single MODEL statement. The first MODEL statement identifies the response variable for the model for COUNT, and it defines a Poisson model with intercept and gender-specific slopes.

The second MODEL statement uses the continuation operator (“+”) and adds a model with a degenerate distribution by using the DIST=CONSTANT option. Because the mass of the constant is placed by default at zero, the second MODEL statement adds a zero-inflation component to the model. When you use the “+” sign, you pick up the response variable and effects from the preceding MODEL statement; any additional effects you list after the “+” sign will be added to them.

Figure 6 displays the “Model Information” and “Optimization Information” tables for this run of the FMM procedure. The model is now identified as a zero-inflated Poisson (ZIP) model with two components, and the parameters continue to be estimated by maximum likelihood. The “Optimization Information” table shows that there are four parameters in the optimization (compared to three parameters in the Poisson GLM model). The four parameters correspond to three parameters in the mean function (intercept and two gender-specific slopes) and the mixing probability.

Figure 7 displays the results from fitting the ZIP model by maximum likelihood. The $-2 \log$ likelihood and the information criteria suggest a much-improved fit over the single-component Poisson model (compare Figure 7 to Figure 5). The Pearson statistic is reduced by a factor of 2 compared to the Poisson model and suggests a better fit than the standard Poisson model.
The number of effective parameters and components shown in Figure 5 equals the values from Figure 6. This is not always the case because components can collapse (for example, when the mixing probability approaches zero or when two components have identical parameter estimates). In this example, both components and all four parameters are identifiable. The Poisson regression and the zero process mix, with a probability of approximately 0.6972 attributed to the Poisson component.

Bayesian Techniques in the FMM Procedure

The FMM procedure enables you to fit some mixture models by Bayesian techniques. The following statements add the BAYES statement to the previous PROC FMM statements:

```sas
proc fmm data=catch seed=12345;
class gender;
model count = gender*age / dist=Poisson;
model + / dist=constant;
performance cpucount=2;
bayes;
run;
```

The “Model Information” table indicates that the model parameters are estimated by Markov chain Monte Carlo techniques, and it displays the random number seed (Figure 8). This is useful if you did not specify a seed to identify the seed value that reproduces the current analysis. The “Bayes Information” table provides basic information about the Monte Carlo sampling scheme. The sampling method uses a data augmentation scheme to impute component membership and then the Gamerman (1997) algorithm to sample the component-specific parameters. The 2,000 burn-in samples are followed by 10,000 Monte Carlo samples without thinning.
The “Prior Distributions” table identifies the prior distributions, their parameters for the sampled quantities, and their initial values. The prior distribution of parameters associated with model effects is a normal distribution with mean 0 and variance 1,000. The prior distribution for the mixing probability is a Dirichlet(1,1), which is identical to a uniform distribution (Figure 8). Since the second mixture component is a degeneracy at zero with no associated parameters, it does not appear in the “Prior Distributions” table in Figure 8.

Figure 9 displays descriptive statistics about the 10,000 posterior samples. Recall from Figure 7 that the maximum likelihood estimates were \(-3.5215, 0.1216, 0.1056,\) and \(0.6972\), respectively. With this choice of prior, the means of the posterior samples are generally close to the MLEs. The “Posterior Intervals” table displays 95% intervals of equal-tail probability and 95% intervals of highest posterior density (HPD) intervals.

**Figure 9** Posterior Summaries and Intervals in the ZIP Model

<table>
<thead>
<tr>
<th>Component</th>
<th>Effect</th>
<th>gender</th>
<th>N</th>
<th>Mean</th>
<th>Deviation</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>Intercept</td>
<td></td>
<td>10000</td>
<td>-3.5524</td>
<td>0.6509</td>
</tr>
<tr>
<td>1</td>
<td>age*gender</td>
<td>F</td>
<td>10000</td>
<td>0.1220</td>
<td>0.0136</td>
</tr>
<tr>
<td>1</td>
<td>age*gender</td>
<td>M</td>
<td>10000</td>
<td>0.1058</td>
<td>0.0140</td>
</tr>
<tr>
<td>1</td>
<td>Probability</td>
<td></td>
<td>10000</td>
<td>0.6938</td>
<td>0.0945</td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th>Component</th>
<th>Effect</th>
<th>gender</th>
<th>25%</th>
<th>50%</th>
<th>75%</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>Intercept</td>
<td></td>
<td>-3.9922</td>
<td>-3.5359</td>
<td>-3.0875</td>
</tr>
<tr>
<td>1</td>
<td>age*gender</td>
<td>F</td>
<td>0.1124</td>
<td>0.1218</td>
<td>0.1314</td>
</tr>
<tr>
<td>1</td>
<td>age*gender</td>
<td>M</td>
<td>0.0961</td>
<td>0.1055</td>
<td>0.1153</td>
</tr>
<tr>
<td>1</td>
<td>Probability</td>
<td></td>
<td>0.6293</td>
<td>0.6978</td>
<td>0.7605</td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th>Component</th>
<th>Effect</th>
<th>gender</th>
<th>Alpha</th>
<th>Equal-Tail Interval</th>
<th>HPD Interval</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>Intercept</td>
<td></td>
<td>0.050</td>
<td>-4.8693 -2.3222 -4.8927</td>
<td>-2.3464</td>
</tr>
<tr>
<td>1</td>
<td>age*gender</td>
<td>F</td>
<td>0.050</td>
<td>0.0960 0.1494 0.0961 0.1494</td>
<td></td>
</tr>
<tr>
<td>1</td>
<td>age*gender</td>
<td>M</td>
<td>0.050</td>
<td>0.0792 0.1339 0.0796 0.1341</td>
<td></td>
</tr>
<tr>
<td>1</td>
<td>Probability</td>
<td></td>
<td>0.050</td>
<td>0.5041 0.8688 0.5025 0.8666</td>
<td></td>
</tr>
</tbody>
</table>

You can generate trace plots for the posterior parameter estimates by enabling ODS Graphics:

```r
ods graphics on;
ods select TADPanel;
proc fmm data=catch seed=12345;
class gender;
model count = gender*age / dist=Poisson;
model + / dist=constant;
performance cpucount=2;
bayes;
run;
ods graphics off;
```

A separate trace panel is produced for each sampled parameter, and the panels for the gender-specific slopes are shown in Figure 10. There is good mixing in the chains: the modest autocorrelation diminishes after about 10 successive samples. By default, the FMM procedure transfers the credible intervals for each parameter from the “Posterior Intervals” table to the trace plot and the density plot in the trace panel.
New Confidence Intervals for the Difference of Proportions

It is often useful to describe the nature of the association in a $2 \times 2$ table by examining the difference of proportions with the ‘event’ response in each of the groups. For example, for the table in Table 2 you would be interested in looking at the difference in the favorable response for the two treatment groups, Test and Control.

<table>
<thead>
<tr>
<th>Treatment</th>
<th>Favorable</th>
<th>Unfavorable</th>
<th>Total</th>
</tr>
</thead>
<tbody>
<tr>
<td>Test</td>
<td>10</td>
<td>2</td>
<td>12</td>
</tr>
<tr>
<td>Control</td>
<td>2</td>
<td>4</td>
<td>6</td>
</tr>
<tr>
<td>Total</td>
<td>12</td>
<td>6</td>
<td>18</td>
</tr>
</tbody>
</table>
The proportion $p_i$ of favorable events in the $i$th row is defined as $n_{i1}/n_{i+}$. When the groups represent simple random samples, and the difference $d = p_1 - p_2$ for the proportions $p_1$ and $p_2$,

$$E\{p_1 - p_2\} = \pi_1 - \pi_2$$

The variance is

$$V\{p_1 - p_2\} = \frac{\pi_1(1-\pi_1)}{n_{1+}} + \frac{\pi_2(1-\pi_2)}{n_{2+}}$$

for which a consistent estimate is

$$\hat{v}_d = \frac{p_1(1-p_1)}{n_{1+}} + \frac{p_2(1-p_2)}{n_{2+}}$$

A $100(1-\alpha)%$ confidence interval for $(\pi_1 - \pi_2)$ with continuity correction is written as

$$d \pm \left\{ z_{\alpha/2} \sqrt{\hat{v}_d + \frac{1}{2} \left\{ \frac{1}{n_{1+}} + \frac{1}{n_{2+}} \right\}} \right\}$$

where $z_{\alpha/2}$ is the $100(1-\alpha/2)$ percentile of the standard normal distribution; this confidence interval is based on Fleiss, Levin, and Paik (2003). These confidence limits include a continuity adjustment to the Wald asymptotic confidence limits that adjust for the difference between the normal approximation and the discrete binomial distribution.

The default Wald asymptotic confidence limits (without the continuity correction) in the FREQ procedure are appropriate for larger sample sizes, say cell counts of at least 12. The continuity-corrected Wald confidence limits are more appropriate for moderate sample sizes, say cell counts of at least 8.

Since the cell counts are relatively small in Table 2, you might consider exact methods for not only assessing association (Fisher’s exact test) but also for the construction of the confidence intervals for the difference of proportions. The EXACT statement provides an exact confidence interval based on the raw risk difference. You specify the RISKDIFF option in the EXACT statement to generate it. However, this interval is known to be fairly conservative (wide). New in SAS/STAT 9.3 is a method for computing the exact confidence interval based on the Farrington-Manning score statistic which has better coverage properties.

Also new in SAS/STAT 9.3 are several other methods for constructing the confidence interval for the difference of proportions: Newcombe, Farrington-Manning, and Hauck-Anderson. These methods can be more suitable for small counts than the exact intervals. See Stokes and Koch (2011) for more detail, as well as the PROC FREQ documentation. The following example produces these new confidence intervals with the FREQ procedure.

The following SAS statements create the SAS data set SMALL:

```sas
data small;
  input treat $ outcome $ count @@ ;
datalines;
Test Favorable 10 Test Unfavorable 2
Placebo Favorable 2 Placebo Unfavorable 4
;
```

The PROC FREQ statements request the new confidence intervals as suboptions of the RISKDIFF option in the TABLES statement. The EXACT statement, with RISKDIFF=(FMSCORE), specifies the exact confidence interval based on the Farrington-Manning score statistic.

```sas
proc freq order=data data=small;
  weight count;
  tables treat*outcome / riskdiff(cl=(wald newcombe fm ha exact) correct);
  exact riskdiff(fmscore);
  ods output PdiffCLs=pd;
  ods output RiskDiffCol1=dif(keep=risk row where=(row='Difference'));
run;
```

The ODS OUTPUT statements are needed in order to plot the intervals in a later step.

Figure 11 displays the generated confidence intervals.
Figure 11  Confidence Intervals for Difference of Proportions

<table>
<thead>
<tr>
<th>Type</th>
<th>95% Confidence Limits</th>
</tr>
</thead>
<tbody>
<tr>
<td>Exact (FM Score)</td>
<td>-0.0000 0.8433</td>
</tr>
<tr>
<td>Farrington-Manning</td>
<td>0.0380 0.9620</td>
</tr>
<tr>
<td>Hauck-Anderson</td>
<td>-0.0516 1.0000</td>
</tr>
<tr>
<td>Newcombe Score (Corrected)</td>
<td>-0.0352 0.8059</td>
</tr>
<tr>
<td>Wald (Corrected)</td>
<td>-0.0571 1.0000</td>
</tr>
</tbody>
</table>

Note that the corrected Wald-based confidence interval is the widest interval. The scored-based exact (unconditional) confidence interval has a relatively wide interval, (−0.0296, 0.8813) as does the interval based on Farrington-Manning scores, (0.0380, 0.9620), which also excludes the value zero. The score-based exact confidence interval has a narrower interval of (0.000, 0.8433), but it comes with a zero boundary. Not shown here, the default exact unconditional interval is (−0.0296, 0.8813), which is fairly wide. The corrected Newcombe interval is the narrowest at (−0.0352, 0.8059), and it might be the most suitable for these data as it does not exclude zero and thus is in harmony with the Fisher exact test result.

By using the SGPLOT procedure, you can easily compare these confidence intervals graphically.

```sas
data risk;
  set pd;
  if _n_ = 1 then set dif(drop=row);
  y = -_n_;
  l = y + 0.25;
run;
ods graphics on;
proc sgplot data=risk noautolegend;
  title 'Confidence Intervals for the Difference of Proportions';
  scatter x = risk y = y / xerrorlower=lowercl xerrorupper=uppercl;
  scatter x = risk y = l / markerchar=type;
  yaxis display=none;
  xaxis offsetmin=0.025 offsetmax=0.025;
run;
ods graphics off;
```

Figure 12 displays the resulting graph.
Frailty Models in Survival Analysis

When experimental units are clustered, the failure times of those units within a cluster tend to be correlated. One approach to managing this correlation when you want to apply Cox regression is to estimate the regression parameters by the maximum partial likelihood estimates, assuming an independent working assumption and accounting for the correlation with a robust sandwich covariance matrix estimate (Lee, Wei, and Amato 1992).

Another approach is to account for within-cluster correlation by using a shared frailty mode in which the cluster effects are incorporated in the model as normally distributed random variables.

The hazard rate for the \( j \)th individual in the \( i \)th cluster is

\[
\lambda_{ij}(t) = \lambda_0(t)e^{\beta \mathbf{Z}_{ij}(t) + \gamma_i}
\]

where \( \lambda_0(t) \) is an arbitrary baseline hazard rate, \( \mathbf{Z}_{ij} \) is the vector of (fixed-effect) covariates, \( \beta \) is the vector of regression coefficients, and \( \gamma_i \) is the random effect for cluster \( i \). The random components \( \gamma_1, \ldots, \gamma_s \) are assumed to be independent and identically distributed as a normal random variable with mean 0 and an unknown variance \( \theta \).

In terms of the frailties \( u_1, \ldots, u_s \), given by \( \gamma_i = \log(u_i) \), the frailty model can be written as

\[
\lambda_{ij}(t) = \lambda_0(t)u_{i}e^{\beta \mathbf{Z}_{ij}(t)}
\]

Each frailty has a lognormal distribution with median 1. This gives the interpretation that individuals in cluster \( i \) with \( u_i > 1 \) (\( u_i < 1 \)) tend to fall at a faster (slower) rate than that under an independence model. The PHREG procedure uses a penalized partial likelihood approach to fit frailty models.

The following DATA step creates the data set BLIND that represents 197 diabetic patients who have a high risk of experiencing blindness in both eyes. One eye of each patient is treated with laser photocoagulation, and the other eye is treated with standard remedies. The hypothesis of interest is whether the laser treatment delays the occurrence of blindness. Since juvenile and adult diabetes have very different courses, it is also desirable to examine how the age of onset of diabetes might affect the time of blindness. Since there are no biological differences between the left eye and the right eye, it is natural to assume a common baseline hazard function for the failure times of the left and right eyes.

Each patient is a cluster that contributes two observations to the input data set, one for each eye. The following variables are in the input data set BLIND:

- ID, patient’s identification
- Time, time to blindness

![Figure 12](image.png)

**Figure 12** Confidence Intervals for Difference of Proportions width
Status, blindness indicator (0:censored and 1:blind)

Treat, treatment received (Laser or Others)

Type, type of diabetes (Juvenile: onset at age \( \leq 20 \) or Adult: onset at age > 20)

The following SAS statements create the data set BLIND:

```sas
proc format;
  value type 0='Juvenile' 1='Adult';
  value Rx 1='Laser' 0='Others';
run;
data Blind;
  input ID Time Status dty trt @@;
  Type= put(dty, type.);
  Treat= put(trt, Rx.);
  datalines;
  5 46.23 0 1 1 5 46.23 0 1 0 14 42.50 0 0 1 14 31.30 1 0 0
  16 42.27 0 0 1 16 42.27 0 0 0 25 20.60 0 0 1 25 20.60 0 0 0
  29 38.77 0 0 1 29 0.30 1 0 0 46 65.23 0 0 1 46 54.27 1 0 0
  49 63.50 0 0 1 49 10.80 1 0 0 56 23.17 0 0 1 56 23.17 0 0 0
  61 1.47 0 0 1 61 1.47 0 0 0 71 58.07 0 1 1 71 13.83 1 1 0
  100 46.43 1 1 1 100 48.53 0 1 0 112 44.40 0 1 1 112 7.90 1 1 0
  ...

The next PROC PHREG statements fit a shared frailty model to the BLIND data set. The RANDOM statement specifies the analysis, and it identifies ID as the clustering variable. The variable ID must also be listed in the CLASS statement.

```sas
proc phreg data=Blind;
  class ID Treat Type;
  model Time*Status(0)=Treat|Type;
  random ID;
  hazardratio 'Frailty Model Analysis' Treat;
run;
```

Figure 13 shows a tabulation of the data. 54 eyes treated with laser photocoagulation developed blindness, as did 101 eyes treated with other remedies.

**Figure 13** Crossstabulations

<table>
<thead>
<tr>
<th>Treat</th>
<th>Status</th>
<th>Frequency</th>
<th>0</th>
<th>1</th>
<th>Total</th>
</tr>
</thead>
<tbody>
<tr>
<td>Laser</td>
<td></td>
<td>143</td>
<td>54</td>
<td>197</td>
<td></td>
</tr>
<tr>
<td>Others</td>
<td></td>
<td>96</td>
<td>101</td>
<td>197</td>
<td></td>
</tr>
<tr>
<td>Total</td>
<td></td>
<td>239</td>
<td>155</td>
<td>394</td>
<td></td>
</tr>
</tbody>
</table>

Figure 14 displays the estimate and asymptotic estimated standard error of the common variance parameter of the normal random effects.

**Figure 14** Variance Estimate of the Normal Random Effects

<table>
<thead>
<tr>
<th>Cov Parm</th>
<th>REML Estimate</th>
<th>Standard Error</th>
</tr>
</thead>
<tbody>
<tr>
<td>ID</td>
<td>0.8308</td>
<td>0.2145</td>
</tr>
</tbody>
</table>
Table 15 displays the Wald tests for both the fixed effects and the random effects. The random effects are statistically significant ($p=0.0042$). Laser photocoagulation appears to be effective ($p=0.0252$) in delaying the occurrence of blindness, although there is also a significant treatment by diabetes type interaction effect ($p=0.0071$).

**Figure 15 Type 3 Tests**

<table>
<thead>
<tr>
<th>Effect</th>
<th>Wald Chi-Square</th>
<th>DF</th>
<th>Pr &gt; ChiSq</th>
<th>Adjusted DF</th>
<th>Pr &gt; ChiSq</th>
</tr>
</thead>
<tbody>
<tr>
<td>Treat</td>
<td>4.8964</td>
<td>1</td>
<td>0.0269</td>
<td>0.9587</td>
<td>0.0252</td>
</tr>
<tr>
<td>Type</td>
<td>2.6386</td>
<td>1</td>
<td>0.1043</td>
<td>0.6795</td>
<td>0.0629</td>
</tr>
<tr>
<td>Treat*Type</td>
<td>7.1336</td>
<td>1</td>
<td>0.0076</td>
<td>0.9644</td>
<td>0.0071</td>
</tr>
<tr>
<td>ID</td>
<td>110.3916</td>
<td>.</td>
<td>.</td>
<td>74.2776</td>
<td>0.0042</td>
</tr>
</tbody>
</table>

**Figure 16 Parameter Estimates**

**Analysis of Maximum Likelihood Estimates**

<table>
<thead>
<tr>
<th>Parameter</th>
<th>Standard Parameter DF Estimate</th>
<th>Error</th>
<th>Chi-Square</th>
<th>Pr &gt; ChiSq</th>
</tr>
</thead>
<tbody>
<tr>
<td>Treat Laser</td>
<td>-0.49849</td>
<td>0.22528</td>
<td>4.8964</td>
<td>0.0269</td>
</tr>
<tr>
<td>Type Adult</td>
<td>0.39781</td>
<td>0.24490</td>
<td>2.6386</td>
<td>0.1043</td>
</tr>
<tr>
<td>Treat*Type Laser Adult</td>
<td>-0.96530</td>
<td>0.36142</td>
<td>7.1336</td>
<td>0.0076</td>
</tr>
</tbody>
</table>

**Analysis of Maximum Likelihood Estimates**

<table>
<thead>
<tr>
<th>Hazard Parameter</th>
<th>Ratio Label</th>
</tr>
</thead>
<tbody>
<tr>
<td>Treat Laser</td>
<td>Treat Laser</td>
</tr>
<tr>
<td>Type Adult</td>
<td>Type Adult</td>
</tr>
<tr>
<td>Treat*Type Laser</td>
<td>Treat Laser * Type Adult</td>
</tr>
</tbody>
</table>

**Figure 17 Hazard Ratio Estimates for Frailty Model**

<table>
<thead>
<tr>
<th>Description</th>
<th>Hazard Ratio</th>
<th>95% Wald Confidence</th>
</tr>
</thead>
<tbody>
<tr>
<td>Treat Laser vs Others At Type=Adult</td>
<td>0.231</td>
<td>0.133</td>
</tr>
<tr>
<td>Treat Laser vs Others At Type=Juvenile</td>
<td>0.607</td>
<td>0.391</td>
</tr>
</tbody>
</table>

These results are similar to those found in a marginal model analysis, which are described in the PHREG documentation.

**Bayesian Software Updates**

Bayesian capabilities continue to grow in SAS/STAT software. First available via web-downloadable procedures, the capabilities provided by the BAYES statement in the GENMOD, LIFEREG, and PHREG procedures have been updated with new sampling methods. Conjugate sampling for the linear regression in the GENMOD procedure is now the default, which provides computation time reductions. In addition, you can specify either the Gamerman algorithm or the independent Metropolis algorithm with the SAMPLING= option in the BAYES statement in PROC GENMOD. You can choose the random-walk Metropolis algorithm as an alternative sampling method in the PHREG procedure. PROC PHREG also added the Zellner $g$-prior for the regression coefficients.

The MCMC procedure introduces the RANDOM statement in this release. It simplifies the construction of hierarchical random-effects models and significantly reduces simulation time while improving convergence, especially in models
with a large number of subjects or clusters. This statement defines random effects that can enter the model in a linear or nonlinear fashion and supports univariate and multivariate prior distributions.

Consider a scenario in which data are collected in groups and you want to model group-specific effects. You can use a random-effects model (sometimes also known as a variance-components model):

\[ y_{ij} = \beta_0 + \beta_1 x_{ij} + \gamma_i + e_{ij}, \quad e_{ij} \sim \text{normal}(0, \sigma^2) \]

where \( i = 1, 2, \ldots, I \) is the group index and \( j = 1, 2, \ldots, n_i \) indexes the observations in the \( i \)th group. In the regression model, the fixed effects \( \beta_0 \) and \( \beta_1 \) are the intercept and the coefficient for variable \( x_{ij} \), respectively. The random effect \( \gamma_i \) is the mean for the \( i \)th group, and \( e_{ij} \) are the error term.

Consider the following SAS data set:

```
title 'Random-Effects Model';
data heights;
  input Family Gender$ Height @@;
datalines;
1 F 67 1 F 66 1 F 64 1 M 71 1 M 72 2 F 63
2 F 63 2 F 67 2 M 69 2 M 68 2 M 70 3 F 63
3 M 64 4 F 67 4 F 66 4 M 67 4 M 67 4 M 69
;```

Height, gender, and family are recorded for 18 individuals. Since GENDER is a character variable and PROC MCMC does not support a CLASS statement, you need to create the corresponding design matrix columns. The next DATA step creates a dummy variable IGender. In this example, the design matrix for a factor variable of level 2 (M and F) can be constructed using the following statement:

```
data iheights;
  set heights;
  IGender=('F');
run;```

To model the data, you can assume that HEIGHT is normally distributed:

\[ y_{ij} \sim \text{normal}(\mu_{ij}, \sigma^2) \]

\[ \mu_{ij} = \beta_0 + \beta_1 g_{ij} + \gamma_i \]

The priors on the parameters \( \beta_0, \beta_1, \gamma_i \) are also assumed to be normal:

\[ \beta_0 \sim \text{normal}(0, \text{var} = 1e5) \]

\[ \beta_1 \sim \text{normal}(0, \text{var} = 1e5) \]

\[ \gamma_i \sim \text{normal}(0, \text{var} = \sigma^2_{\gamma}) \]

Priors on the variance terms, \( \sigma^2 \) and \( \sigma^2_{\gamma} \), are inverse-gamma:

\[ \sigma^2 \sim \text{igamma}(\text{shape} = 0.01, \text{scale} = 0.01) \]

\[ \sigma^2_{\gamma} \sim \text{igamma}(\text{shape} = 0.01, \text{scale} = 0.01) \]

The inverse-gamma distribution is a conjugate prior for the variance in the normal likelihood and the variance in the prior distribution of the random effect.

The following statements fit a linear random-effects model to the data and produce the output shown in Figure 18:

```
ods graphics on;
proc mcmc data=iheights outpost=postout nmc=50000 thin=5 seed=7893;
  ods select Parameters REparameters PostSummaries PostIntervals tadpanel;
  parms b0 0 b1 0 s2 1 s2g 1;
  prior b: ~ normal(0, var = 1e5);
  prior s: ~ igamma(0.01, scale = 0.01);
  random gamma ~ normal(0, var = s2g) subject=family monitor=(gamma);
  mu = b0 + b1 * gf + gamma;
  model height ~ normal(mu, var = s2);
run;
ods graphics off;```
The PROC MCMC statement specifies the input and output data sets, the simulation size, the thinning rate, and a random number seed. The ODS SELECT statement displays the model parameter and random-effects parameter information tables, summary statistics table, the interval statistics table, and the diagnostics plots.

The PARMS statement lumps all four model parameters in a single block. They are B0 (overall intercept), B1 (main effect for IGENDER), S2 (variance of the likelihood function), and S2G (variance of the random effect). If a random walk Metropolis sampler is the only applicable sampler for all parameters, then these four parameters are updated in a single block. However, since the conjugate updater is used to draw posterior samples of S2 and S2G, PROC MCMC updates these parameters separately (see the BLOCK column in the “Parameters” table in Figure 18).

The PRIOR statements specify priors for all the parameters. The notation b: is a shorthand for all symbols that start with the letter 'b'. In this example, b: includes B0 and B1. Similarly, S: stands for both S2 and S2G.

The RANDOM statement specifies a single random effect to be GAMMA, and it specifies that it has a normal prior centered at 0 with variance S2G. The SUBJECT= argument in the RANDOM statement defines a group index FAMILY in the model, where all observations from the same family should have the same group indicator value. The MONITOR= option outputs analysis for all of the random-effects parameters.

Finally, the MU assignment statement calculates the expected value of HEIGHT in the random-effects model. The MODEL statement specifies the likelihood function for HEIGHT.

The “Parameters” and “Random-Effects Parameters” tables, shown in Figure 18, contain information about the model parameters and the four random-effects parameters.

Figure 18 Model and Random-Effects Parameter Information

<table>
<thead>
<tr>
<th>Block</th>
<th>Parameter</th>
<th>Sampling Method</th>
<th>Initial Value</th>
<th>Prior Distribution</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>s2</td>
<td>Conjugate</td>
<td>1.0000</td>
<td>igamma(0.01, scale = 0.01)</td>
</tr>
<tr>
<td>2</td>
<td>s2g</td>
<td>Conjugate</td>
<td>1.0000</td>
<td>igamma(0.01, scale = 0.01)</td>
</tr>
<tr>
<td>3</td>
<td>b0</td>
<td>N-Metropolis</td>
<td>0</td>
<td>normal(0, var = 10000)</td>
</tr>
<tr>
<td></td>
<td>b1</td>
<td></td>
<td>0</td>
<td>normal(0, var = 10000)</td>
</tr>
</tbody>
</table>

Random Effects Parameters

<table>
<thead>
<tr>
<th>Parameter</th>
<th>Subject</th>
<th>Levels</th>
<th>Prior Distribution</th>
</tr>
</thead>
<tbody>
<tr>
<td>gamma</td>
<td>family</td>
<td>4</td>
<td>normal(0, var = s2g)</td>
</tr>
</tbody>
</table>

The posterior summary and interval statistics for B0 and B1 are shown in Figure 19 and Figure 20:

Figure 19 Posterior Summary Statistics

<table>
<thead>
<tr>
<th>Parameter</th>
<th>N</th>
<th>Mean</th>
<th>Standard Deviation</th>
<th>25%</th>
<th>50%</th>
<th>75%</th>
</tr>
</thead>
<tbody>
<tr>
<td>b0</td>
<td>10000</td>
<td>68.4591</td>
<td>1.2085</td>
<td>67.7939</td>
<td>68.4369</td>
<td>69.0863</td>
</tr>
<tr>
<td>b1</td>
<td>10000</td>
<td>-3.5381</td>
<td>0.9622</td>
<td>-4.1439</td>
<td>-3.5351</td>
<td>-2.9466</td>
</tr>
<tr>
<td>s2</td>
<td>10000</td>
<td>4.1495</td>
<td>1.9089</td>
<td>2.8251</td>
<td>3.7353</td>
<td>4.9854</td>
</tr>
<tr>
<td>s2g</td>
<td>10000</td>
<td>4.8368</td>
<td>17.4110</td>
<td>0.2218</td>
<td>1.2534</td>
<td>4.0669</td>
</tr>
<tr>
<td>gamma_1</td>
<td>10000</td>
<td>0.9374</td>
<td>1.2817</td>
<td>0.0832</td>
<td>0.6802</td>
<td>1.6250</td>
</tr>
<tr>
<td>gamma_2</td>
<td>10000</td>
<td>0.0167</td>
<td>1.1399</td>
<td>-0.4145</td>
<td>0.0325</td>
<td>0.5038</td>
</tr>
<tr>
<td>gamma_3</td>
<td>10000</td>
<td>-1.3313</td>
<td>1.6080</td>
<td>-2.1514</td>
<td>-0.9247</td>
<td>-0.1434</td>
</tr>
<tr>
<td>gamma_4</td>
<td>10000</td>
<td>0.0979</td>
<td>1.1495</td>
<td>-0.3470</td>
<td>0.0537</td>
<td>0.5802</td>
</tr>
</tbody>
</table>

16
Trace plots, autocorrelation plots, and posterior density plots for all the parameters are shown in Figure 20. The mixing looks very reasonable, suggesting convergence.

Figure 21 Plots for $b_1$ and Log of the Posterior Density

From the interval statistics table, you see that both the equal-tail and HPD intervals for B0 are positive, strongly indicating the positive effect of the parameter. On the other hand, both intervals for B1 cover the value zero, indicating that IGENDER does not have a strong impact on predicting HEIGHT in this model.

In addition to the default Metropolis-based algorithms, PROC MCMC now takes advantages of certain forms of conjugacy in the model in order to sample directly from the target conditional distributions. In many situations, the conjugate sampler increases sampling efficiency and provides a substantial reduction in computing time.


More Highlights of SAS/STAT 9.3

A number of other important enhancements are available in this release as well.

The EFFECT statement

The EFFECT statement is now production. This statement is available in the HPMIXED, GLIMMIX, GLMSELECT, LOGISTIC, ORTHOREG, PHREG, PLS, QUANTREG, ROBUSTREG, SURVEYLOGISTIC, and SURVEYREG proce-
dures. The NATURALCUBIC option specifies a natural cubic spline basis for the spline expansion.

**CALIS Procedure**

The following features are now production: METHOD=FIML, mean structure analysis with the COSAN model, extended PATH modeling language that supports the specification of variances or covariances as paths, unnamed free parameter specification in all model, and improved improved RAM model specification.

In addition, PROC CALIS now provides detailed analysis of the missing patterns with the FIML estimation method. With the COVPATTERN= and MEANPATTERN= options, you can specify various standard mean and covariance patterns by using keywords. PROC CALIS then generates the required covariance and mean structures automatically.

**NLIN Procedure**

The NLIN procedure provides several experimental features for diagnosing the nonlinear model fit, including the PLOTS, NLINMEASURES, and BIAS options in the PROC NLIN statement. The NLINMEASURES displays global measures of nonlinearity, and the BIAS option computes Box’s bias statistics for the parameter estimates. The PLOTS option enables you to plot the fitted model, fit diagnostics, tangential and Jacobian leverage, and local influence. You can add the leverage, local influence, and residual diagnostics to the output data set.

![Figure 22](image)

**HPMIXED Procedure**

The HPMIXED procedure provides mixed models analysis for when you have large amounts of data and/or large numbers of class levels that may be beyond the grasp of the MIXED procedure. PROC HPMIXED employs sparse matrix methods and other numerical techniques to fit the required mixed models. The HPMIXED procedure became production with SAS/STAT 9.22.

PROC HPMIXED provides the REPEATED statement, which defines the repeated effect and the residual covariance structure in the mixed model. The AR(1), CS, CSH, UC, UCH, and UN covariance structures are now available with the TYPE= option in the RANDOM statement.
MI Procedure

The experimental FCS statement specifies a multivariate imputation by fully conditional specification (FCS) methods. For data with an arbitrary missing data pattern, these methods enable you to impute missing values for all variables, assuming that a joint distribution for these variables exists. The FCS method requires fewer iterations than the MCMC method.

QUANTREG Procedure

The new QINTERACT option in the TEST statement enables you to test whether any difference exists among the coefficients across quantiles if several quantiles are specified in the MODEL statement. The RANKSCORE option in the TEST statement now supports the tau score function, which is appropriate for non-iid error models.

For Further Information

A good place to start for further information is the “What’s New in SAS/STAT 9.3” chapter in the online documentation when it becomes available. In addition, the Statistics and Operations Focus Area includes substantial information about the statistical products, and can be found at support.sas.com/statistics/. The quarterly e-newsletter for that site is available on its home page. And of course, complete information is available in the online documentation located here: support.sas.com/documentation/onlinedoc/stat/.

References


ACKNOWLEDGMENTS

The authors are grateful to Warren Kuhfeld for his graphics contributions. They are also grateful to Tim Arnold for his documentation programming, as well as Jennifer Waller and Rachael Biel for their support of this project.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author:

Maura Stokes
SAS Institute Inc.
SAS Campus Drive
Cary, NC 27513
maura.stokes@sas.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.